IBM SERIES

StripWare™

Creative Computing
STAR GAZING

by James A. Ingram

Contains two programs for amateur and professional astronomers.

Package is not returnable if plastic wrap is cut or removed.

Softstrip™

COMPUTER READABLE PRINT

Suggested Retail Price
$6.98

- Find out which stars are in your sky tonight.
- Print out a list of star locations.
- Valuable to amateur and professional astronomers.
- Convert celestial coordinates to compass bearings.

Cauzin ©1985
CAUZIN SYSTEMS INC
835 SOUTH MAIN STREET, WATERBURY, CT 06706

IBM SERIES

StripWare™

Creative Computing

STAR GAZING

by James A. Ingram

Contains two programs for amateur and professional astronomers.

Package is not returnable if plastic wrap is cut or removed.

CREATIVE COMPUTING, the oldest of the personal computing magazines, was committed to helping the personal computer user get the most from his computer, whether he used it at home, in the office, or in the classroom.

David Ahl started the magazine in 1974 as an aid to the educators he saw reinventing the wheel as they attempted to integrate computers into their curricula. Less than half a year later, hobbyists had begun to buy and build microcomputers to use in their homes. Once the machines were built, however, they found that there was no software for them. They, too, discovered CREATIVE COMPUTING and its entertaining and educational articles and applications.

As the personal computer industry grew and became more sophisticated, CREATIVE COMPUTING kept pace with articles on the latest technology and unbiased reviews of the newest hardware and software on the market.

Welcome to The Cauzin Family

Welcome to the world of data strips. You now own a piece of a revolutionary technological advancement that bridges the gap between print and electronic media more quickly and easily than ever before.

Each data strip package contains selected highlights from among the best software a publisher has to offer.

We at Cauzin hope you'll enjoy this software package, and believe you'll find data strips to be very practical and useful.

Your comments and ideas on this package or any additional applications you'd like to suggest, would be greatly appreciated. Please send your comments or remarks to:

Cauzin Publishing
835 South Main Street
Waterbury, Conn 06706

STAR GAZING

by James A. Ingram

Whether you're a dyed-in-the-wool amateur astronomer or you can't tell a nebula from a noodle, these two IBM BASIC programs will provide many hours of enjoyment. The first (STARS.BAS) is designed to tell you what stars are visible on a given night and where they are located. The second (FSTARS.BAS) is an interactive version to use with a telescope.

I originally wrote them to help position my small field telescope, which doesn't have a clock-drive mounting system for tracking stars. I discovered later that the programs make it super easy for the novice to locate stars and constellations, too. My ninth-grade students give them a thorough workout every fall!

GETTING YOUR BEARINGS

Star position is best described by using right ascension and declination. These are the astronomer's grid system, similar to latitude and longitude. The trouble the average person has in making any sense of them is due to the fact that he is stationary with reference to the stars, while the earth spins on its axis and revolves around the sun. The effect of this is that we have a slightly different view of the sky every night. Each six months we see the opposite half of the sky.

What we need is a grid system that is fastened to the ground where we are standing. The system of compass bearing and altitude is probably the simplest and best known. North is a compass bearing of zero, while east, south, and west are 90°, 180°, and 270° respectively. Full circle brings you to 360°, at the starting point of zero.

Altitude measures the angular height straight up from the horizon in degrees; the horizon is zero and straight overhead is 90° (called the *zenith*). An object at a compass bearing of 135° and altitude of 45° would be halfway between the horizon and overhead to the south-east.

The trick is to relate these two coordinate systems, one moving and one stationary, one tilted at an angle to the other. For the average mathematics professor, this is a piece of cake, but for the rest of us it poses a long and difficult problem. Enter the computer.

If we load the computer with a list of stars and their coordinates, we can let it decide what is visible on a given night and calculate the bearing and altitude. The first step is to freeze any motion, and for that reason the data is calculated only for a specific date and time of night.

Next, the coordinates are mathematically rotated to your local horizon. Finally, the information is printed in a table. The calculations to perform all of this generate some other information useful to the astronomer and these data are included in the table.

USING THE PROGRAM

In STARS, the first of the two BASIC programs, you are asked to enter the date and time. Enter both numbers as a five-character string such as 01/22 or 18:30. Note that numbers must be two-digits and that the time is entered in 24-hour format. The program will accept any value to the nearest day or minute. Your entries are not checked to see if they are real.

Next, you are asked whether you are on standard time or daylight savings time (since the sky rotates 15° every hour, this can make quite a difference in accuracy). Enter **S** for standard or **D** for daylight. Be sure CAPS LOCK is on.

Last, you are asked for a lower limit of brightness to be included in the table. Astronomers call this *magnitude:* the larger the number, the dimmer the star. The brightest stars in the night sky are usually zero or one, while the dimmest visible with the unaided eye are about five or six. For a table of bright stars only, enter a one or two; if you want everything enter a magnitude of 15.

After you enter the brightness limit, the computer takes over the work and sends a table of stars to your printer. First, it prints a table heading with the date, the number of days that have elapsed from the first day of fall (the *autumnal equinox*), the time in three formats, your location in latitude and longitude, the brightness limit, and the position of due south in right ascension coordinates.

It then begins searching through the data table, testing each item to see if it is above your horizon and if it is bright enough to include in the table. If it is, the program finishes the calculations and adds the item to the printed copy. The final table shows the name of the star, its brightness, its right ascension and declination, and its compass bearing and altitude.

UNDER THE STARS OF NIGHT

With list in hand, head for the nearest open field at the appointed time. Establish where north lies (from the North Star, not by the compass), then measure off the bearing and altitude of the first star on the list. You should be able to find it on the first try.

To measure the angles, you can make an instrument called an *astrolabe* for accuracy, or you can use a system of hand measures. Holding your hand at arm's length against the sky the width of your little finger is about one degree; your pointer, middle, and ring fingers together measure about five degrees; your closed fist measures about 10 degrees, and the distance from your pointer to your little finger when spread apart (until it hurts!) is about 15 degrees.

Using both hands, you can work your way around the horizon and up into the sky to the correct location. Just remember, the sky will be turning slowly while you gaze--after an hour or so stars in the south and overhead will be visibly farther west than your table coordinates show.

```
**************************************************************************
**                        TABLE OF VISIBLE SKY OBJECTS                  **
**     THIS PROGRAM CHECKS THROUGH A MASTER TABLE CONTAINING STARS AND OTHER **
** INTERESTING OBJECTS IN THE NIGHT SKY, AND DETERMINES WHICH ONES ARE VIS-  **
** IBLE AT A SPECIFIC TIME AND DATE. IT THEN CALCULATES THE COMPASS BEARING  **
** AND ALTITUDE ABOVE THE HORIZON FOR THAT TIME. THE TABLE  INCLUDES THE     **
** NAME, MAGNITUDE, RIGHT ASCENSION, DECLINATION, BEARING, AND ALTITUDE.     **
** NOTE THAT TIME MUST BE ENTERED IN 24-HOUR FORMAT (2:00 PM=14:00) AND THAT **
** NUMBERS LESS THAN 10 MUST BE ENTERED AS 2 DIGITS WITH LEADING ZERO (05).  **
**************************************************************************
```

I have included stars, nebulae, clusters, and other objects of interest in the data table, but it may not suit your interests. You may want to enter the coordinates of the centers of major constellations, for instance. In any case, such information should be available at your public library in the astronomy section.

USING FSTARS.BAS

The second program (FSTARS.BAS) is simply a compass bearing and altitude calculator. To use it, enter the date and time as before. To locate a given star, look up its right ascension and declination in a star atlas and enter the values into your computer. The display will show some useful values and the bearing and altitude of the star in question.

You may enter data for several stars without having to reenter the date each time; just respond **Y** to the question ANOTHER STAR (Y/N)? Typing **N** stops execution.

CAUZIN'S CORNER • • •

and now for something slightly different

Both of these programs are written in IBM BASIC, so you must enter BASIC to RUN them. To see how the program works, get into BASIC, LOAD the program, and enter LIST. You'll see all the program lines scroll down the screen. Enter LIST 200 to see just one line, in this case line 200.

If you want a printout of either program, get into BASIC and LOAD the program. Type LLIST and press the ENTER key. It will be useful to have a printout of STARS.BAS to have a list of the stars and coordinates stored at the end of the program.

There are several modifications that you have to make so that the programs know your geographic location. By making these values a permanent part of the program, you don't have to type them in each time you use it. You have to tell the programs your longitude and latitude. Town Hall and local airports are good sources for longitude and latitude.

Also, the programs need to know how many hours away from Greenwich Mean Time your time zone is. This is stored in variable GX. Use these values for the continental United States:

```
Eastern Time: GX = 19
Central Time: GX = 18
Mountain Time: GX = 17
Pacific Time: GX = 16
```

In STARS.BAS, change these lines to your local values:

```
250 LNG = 260.3: REM Your longitude goes here
255 LAT = 41.1: REM Local latitude here
260 GX = 18: REM Hours away from GMT
```

In FSTARS.BAS, the lines to change are:

```
210 LAT = 41.4: REM Put your latitude here
215 LGN = 260.3: REM Local longitude here
225 GX = 18: REM Change local time to GMT
```

There are 91 objects available in STARS.BAS. You may want to shorten the table or add more objects (e.g., constellations, comets, etc.). First, change the number in line 265 to match the new data table length. Then, start adding the DATA lines.

Each sky object is on its own DATA line. Add as many as you want, starting with line 1315. Each object needs four pieces of data. Follow the pattern of the other objects, or this outline:

1. Name of object: character string
2. Magnitude: numeric
3. Right ascension: Character string
   - Two digits (use zero where necessary)
   - Decimal point
   - One or two digits
4. Declination (numeric)

CELESTIAL OBJECT AVAILABILITY TABLE

```
DATE  11/15                                  DAYS FROM SEPT. 22 =  54
LOCAL ZONE TIME  21:45 (GMT =  3: 45)        LOCAL STAR TIME  21: 6
LATITUDE  41.4 DEGREES NORTH                 LONGITUDE  260.3 DEGREES EAST
OBJECTS BRIGHTER THAN MAG 5                  ZENITH AT  0: 39 HOURS RA
```

OBJECT	MAG	R.A.	DECL.	BRNG.	ALT.
VEGA (A-LYRA)	0	18.36	38.8	301.5	24.0
CAPELLA (A-AURIGA)	.1	05.14	46.1	59.9	41.7
RIGEL (B-ORION)	.2	05.13	-8.3	110.8	10.2
BETELGEUSE (A-ORION)	.4	05.54	7.4	91.9	13.4
ALTAIR (A-AQUILA)	.8	19.49	8.8	265.1	18.9
ALDEBARAN (A-TAURUS)	.9	04.34	16.5	98.1	34.2
FOMALHAUT (A-PISCIS AUST)	1.2	22.56	-29.7	203.1	15.0
POLLUX (B-GEMINI)	1.2	07.43	28.1	58.8	7.5
DENEB (A-CYGNUS)	1.3	20.40	45.2	296.1	47.3
CASTOR (A-GEMINI)	1.6	07.33	32	57.4	11.7
BELLATRIX (G-ORION)	1.6	05.24	6.4	97.8	18.3
ELNATH (B-TAURUS)	1.7	05.24	28.6	78.3	31.9
ALNILAM (E-ORION)	1.7	05.35	-1.2	101.7	11.2
MIRFAK (A-PERSEUS)	1.8	03.22	49.8	59.4	60.7
DUBHE (A-URSA MAJ)	1.8	11.2	61.9	11.8	15.2
ALIOTH (E-URSA MAJ)	1.8	12.53	56.1	358.1	7.5
ALNITAK (Z-ORION)	1.8	05.39	-2	101.6	9.9
ALKAID (N-URSA MAJ)	1.9	13.46	49.5	349.2	2.1
ALHENA (G-GEMINI)	1.9	06.36	16.4	78.1	11.4
MENKALINAN (B-AURIGA)	1.9	05.57	45	57.4	34.4
POLARIS (A-URSA MIN)	2	02.3	89.1	0.4	42.2
HAMAL (A-ARIES)	2	02.5	23.3	130.6	65.2
DIPHDA (B-CETUS)	2	00.42	-18.2	179.2	30.4
MIRACH (B-ANDROMEDA)	2	01.8	35.5	142.7	82.8
ALPHERATZ (A-ANDROM)	2.1	00.7	28.9	215.1	75.2
ALGOL (B-PERSEUS)	2.1	03.6	40.8	79.4	63.6
KOCHAB (B-URSA MIN)	2.1	14.51	74.3	350.4	27.8
ALMACH (G-ANDROM)	2.1	02.2	42.2	80.2	74.9
SAIPH (K-ORION)	2.1	05.54	7.4	91.9	13.4
SCHEDAR (A-CASSIOPEIA)	2.1	00.39	56.4	359.9	75.0
G-ANDROM. (OR-BL 10)	2	02.2	42.2	80.2	74.9
B-ORION (RIGEL 9)	0	05.13	-8.2	110.7	10.3
S-ORION (TRI. 11-41)	4	05.37	-2.6	102.4	9.8
A-GEMINI (CASTOR 75)	2	07.32	32	57.5	11.8
Z-URSA MAJ (MIZAR 14)	2.5	13.23	55.2	353.7	7.0
NU-DRACO (DBL 62)	5	17.32	55.2	323.0	24.8
E-LYRA (DBL-DBL 2.7)	5	18.11	39.7	305.4	20.6
B-CYGNI (ALBIREO 34)	3	19.30	27.8	284.4	27.0
G-DELPHIN (YL-GR 10)	4.5	20.45	16	261.3	33.9
NGC869 (PERSEUS CL W)	4.5	02.17	57	36.8	67.9
NGC884 (PERSEUS CL E)	4.5	02.21	57	37.6	67.5
ORION NEB (M42)	4	05.33	-5	104.8	8.9
ANDROM GALAXY (M31)	4.8	00.40	41	160.3	89.6

Reading STARS

The following data strips contain STARS.BAS, a program that tells you which stars are in your viewing sky. We've numbered each data strip in the order it must be read into your computer. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strips, you must be in BASIC to run the program. Enter RUN "STARS.BAS". Operating instructions are in the article and the program is menu-driven. Quit anytime by pressing CONTROL-BREAK.



1 2



STARS

by James A. Ingram
Creative Computing Magazine
June 1983

STARS.BAS

```
100 REM ****************************
*********
105 REM **              STARS.BAS
          **
110 REM **  JAMES A. INGRAM   JAN. 7,
 1981  **
115 REM **  CREATIVE  COMPUTING  MAGA
ZINE    **
120 REM **         JUNE  1983  P. 250
          **
125 REM ****************************
*********
128 REM
130 REM ******** Instructions *******
*
135 CLS:WIDTH 80
140 FOR K=1 TO 79:PRINT "*";:NEXT
145 PRINT
150 PRINT "**";TAB(27);"TABLE OF VISI
BLE SKY OBJECTS";TAB(78)"**"
155 PRINT "**      THIS PROGRAM CHECKS
 THROUGH A MASTER TABLE CONTAINING";
160 PRINT " STARS AND OTHER **"
165 PRINT "** INTERESTING OBJECTS IN
THE NIGHT SKY, AND DETERMINES WHICH";

170 PRINT " ONES ARE VIS-  **"
175 PRINT "** IBLE AT A SPECIFIC TIME
 AND DATE. IT THEN CALCULATES THE ";
180 PRINT "COMPASS BEARING  **"
185 PRINT "** AND ALTITUDE ABOVE THE
HORIZON FOR THAT TIME. THE TABLE  ";
190 PRINT "INCLUDES THE     **"
195 PRINT "** NAME, MAGNITUDE, RIGHT
ASCENSION, DECLINATION, BEARING, AN";

200 PRINT "D ALTITUDE.     **"
205 PRINT "** NOTE THAT TIME MUST BE
ENTERED IN 24-HOUR FORMAT (2:00 PM";
210 PRINT "=14:00) AND THAT **"
215 PRINT "** NUMBERS LESS THAN 10 MU
ST BE ENTERED AS 2 DIGITS WITH LE";
220 PRINT "ADING ZERO (05).  **"
225 FOR K=1 TO 79:PRINT "*";:NEXT
235 PRINT
240 REM
245 REM ******** Initialize Constants
 ********
247 REM
248 REM Put you longitude and latitud
e here
249 REM
250 LNG=260.3
255 LAT=41.4
260 GX=18
265 N=91:REM Number of sky objects av
ailable here
270 REM
275 REM ******** Input Local Values *
*******
280 INPUT "Date (MM/DD) : ",D$
285 INPUT "Time (24-hour clock) (HH:M
M) : ",T$
290 INPUT "Standard Time (S) or Dayli
ght (D) : ",F$
295 INPUT "Maximum Magnitude (min. br
ightness) for Table : ",MG
300 REM
305 REM ********** Calculate Basic Va
lues ********
310 REM ** Decipher Time **
315 TH=VAL(LEFT$(T$,2)):REM Zone Hour
s
320 TM=VAL(RIGHT$(T$,2)):REM Zone Min
utes
325 TZ=TH+TM/60:REM Decimal Zone Time

330 IF F$="D" OR F$="d" THEN GX=GX+1:
REM Correct for Daylight Savings Time

335 GT=TZ-GX:REM Uncorrected Decimal
GMT
340 IF GT<0 THEN GT=GT+24
345 IF GT<24 THEN 360
350   GT=GT-24
355 GOTO 345
360 TL=GT+LNG/360*24:REM Uncorrected
Local Star Time
365 IF TL<24 THEN 380
370   TL=TL-24
375 GOTO 365
380 TL$=STR$(INT(TL))+":"+STR$(INT((T
L-INT(TL))*60)):REM Star Time String
385 GT$=STR$(INT(GT))+":"+STR$(INT((G
T-INT(GT))*60)):REM GMT Time String
390 REM
395 REM ** Calculate Days from Sept.
22 **
400 M=VAL(LEFT$(D$,2)):REM Month
405 DY=VAL(RIGHT$(D$,2)):REM Day
410 IF DY=22 AND M=9 THEN CD=0:GOTO 4
85
415 IF DY>22 AND M=9 THEN CD=DY-22:GO
TO 485
420 IF M=10 THEN CD=8+DY:GOTO 485
425 IF M<9 OR (M=9 AND DY<22) THEN M=
M+12:GOTO 440
430 IF M>10 THEN 440
435 PRINT "IMPROPER DATA.  RE-ENTER."
:GOTO 280
440 R=M-10
445 CD=8
450 FOR X=1 TO R:REM Add up months
455    READ D1
460    CD=CD+D1
465 NEXT X
470 DATA 31,30,31,31,28,31,30,31,30,3
1,31,30
475 RESTORE 860:REM Set data pointer
to first star in table
480 CD=CD+DY:REM Add in days in curre
nt month
485 ZLH=CD*24/365+TL
490 IF ZLH>24 THEN ZLH=ZLH-24
495 ZLH$=STR$(INT(ZLH))+":"+STR$(INT(
(ZLH-INT(ZLH))*60))
500 REM
505 REM ******** Print Table Heading
********
515 LPRINT TAB(22);"CELESTIAL OBJECT
AVAILABILITY TABLE"
520 GOSUB 820:REM Printer Delay
530 LPRINT:LPRINT
535 LPRINT "DATE  ";D$;TAB(50)"DAYS F
ROM SEPT. 22 = ";CD
540 LPRINT "LOCAL ZONE TIME  ";T$;" (
GMT = ";GT$;")";
545 LPRINT TAB(50)"LOCAL STAR TIME ";
TL$
550 GOSUB 820:REM Printer Delay
555 LPRINT "LATITUDE ";LAT;"DEGREES N
ORTH";
560 LPRINT TAB(50)"LONGITUDE ";LNG;"L
EGREES EAST"
565 LPRINT "OBJECTS BRIGHTER THAN MAG
";MG;
570 LPRINT TAB(50)"ZENITH AT ";ZLH$;"
 HOURS RA"
575 GOSUB 820:REM Printer Delay
580 LPRINT:LPRINT:LPRINT
585 LPRINT "OBJECT";TAB(35)"MAG";TAB(
40)"R.A.";
590 LPRINT TAB(50)"DECL.";TAB(60)"BRN
G.";TAB(70)"ALT."
600 FOR K=1 TO 79:LPRINT "*";:NEXT
605 LPRINT
610 GOSUB 820: REM Printer Delay
615 REM
620 REM ******** Main Calculation Loo
p ********
625 LAT=LAT*6.28318/360
630 LAT=3.14159/2-LAT
635 FOR J=1 TO N:REM Read data table
640    READ N$,BR,RA$,D
645    IF BR>MG THEN 800:REM Below br
ightness limit
650       RA=VAL(LEFT$(RA$,2))+VAL(RI
GHT$(RA$,2))/60
655       ZRA=(RA-ZLH)*15:REM Positio
n from zenith line in degrees
660       IF ZRA>360 THEN ZRA=ZRA-360

665       IF ZRA<0 THEN ZRA=ZRA+360
670       ALT=90-D
675       ZPI=ZRA*6.28318/360
680       API=ALT*6.28318/360
685       X=SIN(API)*COS(ZPI):REM Con
vert to XYZ coord.
690       Y=SIN(API)*SIN(ZPI)
695       Z=COS(API)
700       X1=X*COS(LAT)-Z*SIN(LAT):RE
M Rotate to zenith
705       Z1=X*SIN(LAT)+Z*COS(LAT)
710       X=X1:Z=Z1
715 REM NOTE: ATN() functions below a
re valid from -PI/2 to PI/2 only.
720 REM       Altitude is within this
 range, but bearing must be translate
d
725 REM       to compass heading with
in each quadrant of the X-Y plane.
730       A=ATN(Z/(SQR(X*X+Y*Y)))*360
/6.283:REM Altitude above horizon
735       IF A<0 THEN 795:REM Below h
orizon?
740          B=ATN(Y/X)*360/6.283:REM
 Compass bearing
745          IF X>0 AND Y>0 THEN B=18
0-B:GOTO 765:REM SE quadrant
750          IF X>0 AND Y<0 THEN B=18
0-B:GOTO 765:REM SW quadrant
755          IF X<0 AND Y>0 THEN B=-B
:GOTO 765:REM NE quadrant
760          IF X<0 AND Y<0 THEN B=36
0-B:GOTO 765:REM NW quadrant
765          REM ** Print Data **
770             LPRINT N$;TAB(35)BR;T
AB(40)RA$;TAB(50)D;
775             LPRINT TAB(60);USING
"###.#";B;
780             LPRINT TAB(70);USING
"###.#";A
785             GOSUB 840:REM Short P
rinter Delay
790          REM ** End of Print Rout
ine **
```

```
795       REM ** End of Bearing Calcu
lation & Print **
800    REM ** End of Visible Stars Onl
y Routine **
805 NEXT J:REM Get Next Star
815 END
820 REM
825 REM ******** Printer Delay Loops
********
830 FOR Q=1 TO 500:NEXT Q
835 RETURN
840 FOR Q=1 TO 100:NEXT Q
845 RETURN
850 REM
855 REM ******** Data Tables ********

860 DATA SIRIUS (A-CANIS MAJ),-1.6,06
.43,-16.7
865 DATA CANOPUS (A-CARINA),-.7,06.23
,-52.7
870 DATA ARCTURUS (A-BOOTES),-.1,14.1
4,19.4
875 DATA VEGA (A-LYRA),0,18.36,38.8
880 DATA CAPELLA (A-AURIGA),.1,05.14,
46.1
885 DATA RIGEL (B-ORION),.2,05.13,-8.
3
890 DATA PROCYON (A-CANIS MIN),.4,07.
37,5.4
895 DATA BETELGEUSE (A-ORION),.4,05.5
4,7.4
900 DATA ALTAIR (A-AQUILA),.8,19.49,8
.8
905 DATA ALDEBARAN (A-TAURUS),.9,04.3
4,16.5
910 DATA ANTARES (A-SCORPIO),1,16.28,
-26.3
915 DATA SPICA (A-VIRGO),1,13.24,-11
920 DATA FOMALHAUT (A-PISCIS AUST),1.
2,22.56,-29.7
925 DATA POLLUX (B-GEMINI),1.2,07.43,
28.1
930 DATA DENEB (A-CYGNUS),1.3,20.40,4
5.2
935 DATA REGULUS (A-LEO),1.4,10.7,12.
1
940 DATA CASTOR (A-GEMINI),1.6,07.33,
32
945 DATA ADHARA (E-CANIS MAJ),1.6,06.
57,-28.9
950 DATA BELLATRIX (G-ORION),1.6,05.2
4,6.4
955 DATA SHAULA (L-SCORPIO),1.6,17.32
,-37.1
960 DATA ELNATH (B-TAURUS),1.7,05.24,
28.6
965 DATA ALNILAM (E-ORION),1.7,05.35,
-1.2
970 DATA MIRFAK (A-PERSEUS),1.8,03.22
,49.8
975 DATA DUBHE (A-URSA MAJ),1.8,11.2,
61.9
980 DATA ALIOTH (E-URSA MAJ),1.8,12.5
3,56.1
985 DATA GAMMA VELORUM,1.8,08.9,-47.3

990 DATA KAUS AUSTRALIS (E-SAGTRS),1.
8,18.22,-34.4
995 DATA ALNITAK (Z-ORION),1.8,05.39,
-2
1000 DATA AL NAIR (A-GRUS),1.8,22.6,-
47.1
1005 DATA ALKAID (N-URSA MAJ),1.9,13.
46,49.5
1010 DATA ALHENA (G-GEMINI),1.9,06.36
,16.4
1015 DATA WEZEN (D-CANIS MAJ),1.9,07.
7,-26.4
1020 DATA THETA SCORPII,1.9,17.35,-43

1025 DATA MENKALINAN (B-AURIGA),1.9,0
5.57,45
1030 DATA MIRZAM (B-CANIS MAJ),2,06.2
1,-17.9
1035 DATA DELTA VELORUM,2,08.44,-54.6

1040 DATA POLARIS (A-URSA MIN),2,02.3
,89.1
1045 DATA ALPHARD (A-HYDRA),2,09.26,-
8.5
1050 DATA HAMAL (A-ARIES),2,02.5,23.3

1055 DATA DIPHDA (B-CETUS),2,00.42,-1
8.2
1060 DATA MENKENT (TH-CENTAURI),2,14.
5,-36.2
1065 DATA MIRACH (B-ANDROMEDA),2,01.8
,35.5
1070 DATA NUNKI (S-SAGGTRS),2.1,18.53
,-26.3
1075 DATA RASALHAQUE (A-OPHIUC),2.1,1
7.34,12.6
1080 DATA ALPHERATZ (A-ANDROM),2.1,00
.7,28.9
1085 DATA ALGOL (B-PERSEUS),2.1,03.6,
40.8
1090 DATA KOCHAB (B-URSA MIN),2.1,14.
51,74.3
1095 DATA ALMACH (G-ANDROM),2.1,02.2,
42.2
1100 DATA SAIPH (K-ORION),2.1,05.54,7
.4
1105 DATA SCHEDAR (A-CASSIOPEIA),2.1,
00.39,56.4
1110 DATA PSI-PISCES (DBL 30),5.5,01.
4,21.2
1115 DATA G-ANDROM. (OR-BL 10),2,02.2
,42.2
1120 DATA B-ORION (RIGEL 9),0,05.13,-
8.2
1125 DATA TH-ORION (TRAPEZ),5.5,05.34
,-5.5
1130 DATA S-ORION (TRI. 11-41),4,05.3
7,-2.6
1135 DATA A-CANIS MAJ (SIRIUS 10),-1.
5,06.44,-16.7
1140 DATA A-GEMINI (CASTOR 75),2,07.3
2,32
1145 DATA T-CANCER (OR-BL 31),4,08.45
,29
1150 DATA G-LEO (DBL 4.5),2.5,10.18,2
0.2
1155 DATA Z-URSA MAJ (MIZAR 14),2.5,1
3.23,55.2
1160 DATA E-BOOTES (GD-BL 3),2.5,14.4
4,27.3
1165 DATA Z-CORONA (DBL 6),5,15.39,36
.8
1170 DATA NU-DRACO (DBL 62),5,17.32,5
5.2
1175 DATA E-LYRA (DBL-DBL 2.7),5,18.1
1,39.7
1180 DATA B-CYGNI (ALBIREO 34),3,19.3
0,27.8
1185 DATA G-DELPHIN (YL-GR 10),4.5,20
.45,16
1190 DATA NGC869 (PERSEUS CL W),4.5,0
2.17,57
1195 DATA NGC884 (PERSEUS CL E),4.5,0
2.21,57
1200 DATA NGC1528 (60 ST. 25'),6.5,04
.14,51.2
1205 DATA NGC2632 (M44 BEEHIVE),4,08.
39,20.1
1210 DATA NGC6523 (M8 LACOON),5,18.2,
-24.4
1215 DATA NGC6618 (M17 OMEGA),7,18.19
,-16.2
1220 DATA NGC6720 (M57 RING),9,18.53,
33
1225 DATA NGC6853 (M27 DUMBL),7.5,19.
59,22.6
1230 DATA NGC5139 (OMEGA GLOB),4,13.2
5,-47.4
1235 DATA NGC6205 (M13 HERC),5.5,16.4
1,36.5
1240 DATA NGC6656 (M22 SAGTR),6,18.35
,-23.9
1245 DATA NGC1952 (M1 CRAB),8.5,05.33
,22
1250 DATA ORION NEB (M42),4,05.33,-5
1255 DATA ANDROM GALAXY (M31),4.8,00.
40,41
1260 DATA CANES CLUSTER (M3),6.3,13.4
0,29
1265 DATA WHIRLPOOL GAL (M51),8.1,13.
28,47
1270 DATA CYGNUS CLUST (M39),5.2,21.3
0,48
1275 DATA OPHIUC CLUST (M10),6.7,16.5
5,-4
1280 DATA OPHIUC CLUST (M12),6.6,16.4
5,-2
1285 DATA OPHIUC CLUST (M14),5.7,17.3
5,-3
1290 DATA PERSEUS SNGL CLSTR (M34),5.
5,02.39,43
1295 DATA AQUARIUS CLSTR (M2),6.3,21.
31,-1
1300 DATA GEMINI CLSTR (M35),5.3,06.6
,24
1305 DATA CANIS MAJ OC (M41),4.6,06.4
5,-21
1310 DATA MONOCEROS OC (M50),6.3,07.1
,-8
```

Reading FSTARS

The following data strip contains FSTARS.BAS, a program to convert celestial coordinates to compass bearings. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, you must be in BASIC to run the program. Enter RUN "FSTARS.BAS". Operating instructions are in the article and the program is menu-driven. Quit anytime by pressing CONTROL-BREAK.



FSTARS

by James A. Ingram
Creative Computing Magazine
June 1983

FSTARS.BAS

```
100 REM ******************************
*********
105 REM **            FSTARS.BAS
      **
110 REM **  JAMES A. INGRAM  JAN. 14,
 1981  **
115 REM **  CREATIVE  COMPUTING  MAGA
ZINE   **
120 REM **        JUNE  1983  P. 250
      **
125 REM ******************************
*********
128 REM
130 CLS:REM Clear screen
140 FOR K=1 TO 70:PRINT"*";:NEXT
145 PRINT:PRINT
150 PRINT TAB(22)"STARS.BAS - INTERAC
TIVE VERSION"
155 PRINT "     This program determin
es the compass bearing and altitude o
f"
160 PRINT "any star at a specific tim
e and date.  You must enter the date,
"
165 PRINT "the time, and the right as
cension and declination of the star."

170 PRINT "Date, time, and right asce
nsion numbers must be entered as a fi
ve"
175 PRINT "digit string, such as '01/
14' or '04:30'.  Time must be in 24-"

180 PRINT "hour format."
185 PRINT
190 FOR K=1 TO 70:PRINT "*";:NEXT
195 PRINT
200 REM
205 REM ******** Initialize ********
210 LAT=41.4:REM Local latitude
215 LNG=260.3:REM Local longitude
220 ROT=(90-LAT)*6.28318/360:REM Rota
tion from Polaris to Zenith
225 GX=18:REM Correct local zone time
 to GMT
230 REM
235 REM ******** Input current variab
les ********
240 INPUT "Enter Date (MM/DD) : ",D$
250 INPUT "Enter Time in 24-hour Form
at (HH:MM) : ",T$
255 INPUT "Enter Right Ascension (HH:
MM) : ",RA$
260 INPUT "Enter Declination (D.DD) :
 ",D
265 REM
270 REM ** Calculate days from Sept.
22 **
275 M=VAL(LEFT$(D$,2)):REM Month
280 DY=VAL(RIGHT$(D$,2)):REM Day
285 IF DY=22 AND M=9 THEN CD=0:GOTO 3
55
290 IF DY>22 AND M=9 THEN CD=DY-22:GO
TO 355
295 IF M=10 THEN CD=8+DY:GOTO 355
300 IF M<9 OR (M=9 AND DY<22) THEN M=
M+12:GOTO 310
305 IF M>10 THEN 310
310 R=M-10
315 CD=8
320 FOR X=1 TO R:REM Add up months
325    READ D1
330    CD=CD+D1
335 NEXT X
340 DATA 31,30,31,31,28,31,30,31,30,3
1,31,30
345 RESTORE 340
350 CD=CD+DY:REM Add in days in curre
nt month
355 ZLH=CD*24/365+TL
360 IF ZLH>24 THEN ZLH=ZLH-24
365 ZLH$=STR$(INT(ZLH))+":"+STR$(INT(
(ZLH-INT(ZLH))*60))
370 REM ******** Calculate basic valu
es ********
375 REM ** Decipher time **
380 TH=VAL(LEFT$(T$,2)):REM Zone hour
s
385 TM=VAL(RIGHT$(T$,2)):REM Zone min
utes
390 TZ=TH+TM/60:REM Decimal zone time

395 IF F$="D" OR F$="d" THEN GX=GX-1:
REM Correct for Daylight Savings Time

400 GT=TZ-GX:REM Uncorrected decimal
GMT
405 IF GT<0 THEN GT=GT+24
410 IF GT<24 THEN 425
415   GT=GT-24
420 GOTO 410
425 TL=GT+LNG/360*24:REM Uncorrected
local star time
430 IF TL<24 THEN 445
435   TL=TL-24
440 GOTO 430
445 TL$=STR$(INT(TL))+":"+STR$(INT((T
L-INT(TL))*60)):REM Star time string
450 GT$=STR$(INT(GT))+":"+STR$(INT((G
T-INT(GT))*60)):REM GMT string
455 REM End of calculations
460 REM ******** Main calculation loo
p *********
465    RA=VAL(LEFT$(RA$,2))+VAL(RIGHT
$(RA$,2))/60
470    ZRA=(RA-ZLH)*15:REM Position f
rom zenith line in degrees
475    IF ZRA>360 THEN ZRA=ZRA-360
480    IF ZRA<0 THEN ZRA=ZRA+360
485    ALT=90-D
490    ZPI=ZRA*6.28318/360
495    API=ALT*6.28318/360
500    X=SIN(API)*COS(ZPI):REM Conver
t to XYZ coord.
505    Y=SIN(API)*SIN(ZPI)
510    Z=COS(API)
515    X1=X*COS(ROT)-Z*SIN(ROT):REM R
otate to zenith
520    Z1=X*SIN(ROT)+Z*COS(ROT)
525    X=X1:Z=Z1
530 REM NOTE: ATN() functions below a
re valid from -PI/2 to PI/2 only,
535 REM       altitude is within this
 range, but bearing must be
540 REM       translated to compass h
eading within each quadrant of
541 REM       the X-Y plane.
545    A=ATN(Z/(SQR(X*X+Y*Y)))*360/6.
283:REM Altitude above horizon
550    IF A<0 THEN PRINT:PRINT: PRINT
 "Star is below horizon.":GOTO 625
555       B=ATN(Y/X)*360/6.283:REM Co
mpass bearing
560       IF X>0 AND Y>0 THEN B=180-B
:GOTO 580:REM SE quadrant
565       IF X>0 AND Y<0 THEN B=180-B
:GOTO 580:REM SW quadrant
570       IF X<0 AND Y>0 THEN B=-B:GO
TO 580:REM NE quadrant
575       IF X<0 AND Y<0 THEN B=360-B
:GOTO 580:REM NW quadrant
580 REM End of calculations
585 CLS:REM Clear screen
590 PRINT "LOCAL ZONE TIME = ";T$;TAB
(40)"LOCAL STAR TIME = ";TL$
595 PRINT "ZENITH MERIDIAN RA = ";ZLH
$;TAB(40)"DAYS FROM 9/22 = ";CD
600 PRINT "RIGHT ASCENSION = ";RA$;TA
B(40)"DECLINATION =";D
605 PRINT
610 PRINT "COMPASS BEARING = ";USING
"###.#";B;
615 PRINT TAB(40)"ALTITUDE = ";USING
"###.#";A
620 PRINT
625 PRINT "Another star (Y/N) ";
630 K$=INKEY$:IF K$="" THEN 630
635 PRINT K$:IF K$="Y" OR K$="y" THEN
 CLS:GOTO 250
640 END
```

LIMITED WARRANTY

Cauzin StripWare™

835 South Main Street
Waterbury, Connecticut 06706

CAUZIN SYSTEMS, INC. (CAUZIN) warrants this package against defects in materials and workmanship for a period of 90 days from the date of original purchase.

If you discover a defect, CAUZIN will, at its option, either repair, replace, or refund the purchase price of this product at no charge to you, provided it is returned during the warranty period indicated above or any extended period(s) paid for by the user. Transportation charges should be prepaid, to the authorized dealer from whom you purchased it or to any other authorized CAUZIN dealer in the country.

You may obtain additional information from CAUZIN directly at the address printed on this certificate, or by calling (203) 573-0150. Please attach your name, address, telephone number, a description of the problem, and a copy of a bill of sale as proof of date of original purchase, to each product returned for warranty service.

This warranty applies only to products manufactured by or for CAUZIN, which can be identified by the CAUZIN trademark, trade name, or logo affixed to the product. CAUZIN does not warrant any product(s) that are not CAUZIN products. This warranty does not apply if the product has been damaged by accident, abuse, misuse or misapplication, has been modified without the written permission of CAUZIN, or if any CAUZIN serial number has been removed, altered, or defaced.

ALL WARRANTIES, EXPRESSED OR IMPLIED, INCLUDING WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE, ARE LIMITED IN DURATION TO 90 DAYS FROM THE ORIGINAL DATE OF PURCHASE. THERE ARE NO WARRANTIES WHICH EXTEND BEYOND THE DESCRIPTION ON THE FACE HEREOF.

THE WARRANTY AND REMEDIES SET FORTH ABOVE ARE EXCLUSIVE AND IN LIEU OF ALL OTHERS, ORAL OR WRITTEN, EXPRESSED OR IMPLIED. NO CAUZIN DEALER, AGENT, OR EMPLOYEE IS AUTHORIZED TO MAKE ANY MODIFICATION, OR ADDITION TO THIS WARRANTY.

CAUZIN IS NOT RESPONSIBLE FOR SPECIAL, INCIDENTAL, OR CONSEQUENTIAL DAMAGES RESULTING FROM ANY BREACH OF WARRANTY, OR UNDER ANY OTHER LEGAL THEORY, INCLUDING LOST PROFITS, DOWNTIME, GOODWILL, DAMAGE TO, OR REPLACEMENT OF EQUIPMENT AND PROPERTY, AND ANY COST OF RECOVERING, REPROGRAMMING OR REPRODUCING ANY PROGRAM OR DATA STORED IN OR USED WITH CAUZIN PRODUCTS.

Somes states do not allow the exclusion or limitation of incidental or consequential damages or limitations on how long an implied warranty lasts, so the above limitations or exclusion may not apply to you.

This warranty shall not be applicable to the extent that any provision of this warranty is prohibited by an federal, state or municipal law which cannot be preempted. This warranty gives you specific legal rights, and you may also have other rights which vary from state to state.